

Aristotle University of Thessaloniki

Computer Science Department

---

**Semester assignment in NNA-07-08**

*Digital Electronic Systems in VHDL*

---

Alexandros Korkos  
**alexkork@csd.auth.gr**  
3870

---

Thessaloniki, January 16, 2024



---

This work is licensed under a **Create Commons "Attribution - Non Commercial - Share Alike 4.0 International"**.

## Abstract

Digital design of logic circuits is a fundamental part of computer engineering. The design of logic circuits is done using Hardware Description Languages (HDLs). The most common HDLs are Verilog and VHDL. This project focuses on the design of logic circuits using VHDL. The project is divided into two parts. The first part focuses on the design of combinational circuits, while the second part focuses on the design of sequential circuits. All the circuits designed and implemented are fundamental circuits, that are used in the design of more complex circuits.

Besides the design of the circuits, the project also focuses on the simulation and validation of the circuits. All the above are done using free and open source software. The software used for the design and simulation of the circuits are GHDL and GTKWave (see Appendix B). The VHDL codes for the circuits are following the version 2008 of IEEE standard 1076.

All VHDL codes are available at [GitHub](#).

## Contents

<b>I</b>	<b>Sequential circuits</b>	<b>4</b>
1	D flip-flop	4
2	JK flip-flop	6
3	Random Access Memory (RAM)	8
4	Sequential multiplier	10
<b>II</b>	<b>Combinatorial circuits</b>	<b>14</b>
5	LED Counter	14
6	Multiplexer	16
7	Full adder	18
8	Comparator	20
9	Encoder	21
<b>A</b>	<b>Supplementary code for LED counter &amp; multiplier</b>	<b>24</b>
A.1	LED counter . . . . .	24
A.1.1	LED Decoder . . . . .	24
A.1.2	Counter . . . . .	25
A.2	Multiplier . . . . .	25
A.2.1	Adder/Subtractor . . . . .	25
A.2.2	Register . . . . .	26
A.2.3	Parallel Shift Register . . . . .	27
<b>B</b>	<b>Tools used for digital design and VHDL compilation</b>	<b>29</b>

## Part I

# Sequential circuits



A sequential circuit, consists of a combinational circuit to which storage elements are connected to form a feedback path. The storage elements are devices capable of storing binary information. The binary information stored in these elements at any given time defines the state of the sequential circuit at that time. The sequential circuit receives binary information from external inputs that, together with the present state of the storage elements, determine the binary value of the outputs.

## 1 D flip-flop

A D flip-flop, is a type of digital storage element used in digital circuits, particularly in sequential logic systems like registers and memories. The output of the D flip-flop follows the value of the D input when the appropriate synchronization pulses are applied.

In this implementation, the D flip-flop is triggered on the rising edge of the clock signal. Its operation is shown in the following truth table.

**Table 1:** D flip-flop truth table

clk	D	$Q_{t+1}$
	0	0
	1	1
other	X	$Q_t$

In the implementation seen in [listing 1](#) the negated Q is not configured, because it was not required by the instructor. Also, the [listing 1](#) uses an asynchronous reset.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3
4  entity d_flip_flop is
5      port (
6          clk          : in  std_logic;
7          rst_n, preset : in  std_logic;
8          D_in         : in  std_logic;
9          Q_out        : out std_logic
10     );
11 end entity;
12
13 -- asynchronous reset
14 architecture behavior of d_flip_flop is
15 begin
16     DUT: process(clk, rst_n, preset)
17     begin
18         if (rst_n = '0') then
19             Q_out ≤ '0';
20         elsif (preset = '0') then
21             Q_out ≤ '1';
22         elsif (clk'event and clk = '1') then
23             Q_out ≤ D_in;
24         end if;
25     end process;
26 end architecture;
```

**Listing 1:** D flip-flop

For the testing of the circuit the test bench created as shown here [listing 2](#) finishes in 90ns without any errors. Additionally, it uses the asserted instruction, to check instantly the value of the output. The clock period simulated is 50ns long, where each 25ns edge is rising/falling.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use std.env.finish;
4
5  entity tb_d_flip_flop is
6  end entity;
7
8  architecture bench of tb_d_flip_flop is
9      signal D_tb_in: std_logic := '0';
10     signal clk_tb: std_logic := '0';
11     signal rst_n_tb: std_logic := '0';
12     signal preset_tb: std_logic := '0';
13     signal Q_tb_out: std_logic;
14
15     constant clk_period : time := 50 ns;
16
17 begin
18     UUT: entity work.d_flip_flop port map (clk_tb, rst_n_tb, preset_tb, D_tb_in, Q_tb_out
19         ↪ );
20
21     clk_process: process
22     begin
23         clk_tb ≤ '0';
24         wait for clk_period/2;
25         clk_tb ≤ '1';
26         wait for clk_period/2;
27     end process;
28
29     stimulus : process
30     begin
31         rst_n_tb ≤ '0';
32         wait for 30 ns;
33         assert(Q_tb_out = '0') report "Error in q" severity failure;
34
35         rst_n_tb ≤ '1';
36         preset_tb ≤ '1';
37         D_tb_in ≤ '0';
38         wait for 15 ns;
39         assert(Q_tb_out = '0') report "Error in q" severity failure;
40
41         rst_n_tb ≤ '1';
42         preset_tb ≤ '1';
43         wait for 15 ns;
44         assert(Q_tb_out = '0') report "Error in q" severity failure;
45
46         rst_n_tb ≤ '1';
47         preset_tb ≤ '0';
48         wait for 10 ns;
49         assert(Q_tb_out = '1') report "Error in q" severity failure;
50
51         rst_n_tb ≤ '1';
52         preset_tb ≤ '1';
53         wait for 20 ns;
54         assert(Q_tb_out = '0') report "Error in q" severity failure;
55
56     finish;
57 end process;
58 end architecture;

```

**Listing 2:** Testbench for D flip-flop

The waveform of this circuit simulation is shown in the [fig. 1](#) following.




Figure 1: Waveform of D flip-flop

## 2 JK flip-flop

A JK flip-flop is a type of digital storage element that has the ability to toggle its output between two states. It is a synchronous device, meaning that it only responds to input signals at transitions of a clock signal. The JK flip-flop is a refinement of the SR flip-flop, where S and R inputs are combined into a single input J (set) with the addition of a clock input and a second input K (reset). The output of the JK flip-flop toggles between two states according to the inputs.

In this implementation, the JK flip-flop is triggered on the rising edge of the clock signal. Its operation is shown in the following truth table 2.

Table 2: JK flip-flop truth table

clk	J	K	$Q_{t+1}$
0	0	0	$Q_t$
0	0	1	0
0	1	0	1
0	1	1	$\overline{Q_t}$
other	X	X	$Q_t$

In the implementation seen in listing 3 the negated Q output is like in the D flip-flop implementation, not configured. Also, the listing 3 uses an asynchronous reset.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3
4  entity jk_flip_flop is
5      port (
6          J_in: in  std_logic;
7          K_in: in  std_logic;
8          rst, clk: in  std_logic;
9          Q_out: inout std_logic
10     );
11 end entity;
12
13 architecture behavior of jk_flip_flop is
14 begin
15     DUT: process (rst, clk) is
16     begin
17         if rst = '1' then
18             Q_out ≤ '0';
19         elsif(rising_edge(clk)) then
20             if (J_in = '0' and K_in = '0') then
21                 Q_out ≤ Q_out;
22             elsif (J_in = '0' and K_in = '1') then
23                 Q_out ≤ '0';
24             elsif (J_in = '1' and K_in = '0') then
25                 Q_out ≤ '1';
26             elsif (J_in = '1' and K_in = '1') then
27                 Q_out ≤ not (Q_out);
28             end if;
29         end if;
30     end process;
31 end behavior;

```

**Listing 3:** JK flip-flop

For the testing of this JK flip-flop, the test bench created as shown here [listing 4](#) finishes in 100ns without any errors. Additionally, it uses the asserted instruction, to check instantly the value of the output. Here the clock period simulated is 10ns long, where each 5ns edge is rising/falling.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use std.env.finish;
4
5  entity tb_jk_flip_flop is
6  end entity;
7
8  architecture bench of tb_jk_flip_flop is
9      signal J_tb_in: std_logic := '0';
10     signal K_tb_in: std_logic := '0';
11     signal rst_n_tb: std_logic := '0';
12     signal clk_tb: std_logic := '0';
13     signal Q_tb_out: std_logic;
14
15     constant clk_period : time := 10 ns;
16
17     begin
18         UUT: entity work.jk_flip_flop port map (J_tb_in, K_tb_in, rst_n_tb, clk_tb, Q_tb_out
19             ↪ );
20
21         clk_process : process
22         begin
23             clk_tb ≤ '0';
24             wait for clk_period/2;
25             clk_tb ≤ '1';
26             wait for clk_period/2;
27         end process;
28
29         stimulus: process
30         begin
31             J_tb_in ≤ '1';
32             K_tb_in ≤ '0';
33             rst_n_tb ≤ '0';
34             wait for 20 ns;
35             assert(Q_tb_out = '1') report "Error in q" severity failure;
36
37             J_tb_in ≤ '0';
38             K_tb_in ≤ '0';
39             rst_n_tb ≤ '0';
40             wait for 20 ns;
41             assert(Q_tb_out = '1') report "Error in q" severity failure;
42
43             J_tb_in ≤ '0';
44             K_tb_in ≤ '1';
45             rst_n_tb ≤ '0';
46             wait for 20 ns;
47             assert(Q_tb_out = '0') report "Error in q" severity failure;
48
49             J_tb_in ≤ '1';
50             K_tb_in ≤ '1';
51             rst_n_tb ≤ '0';
52             wait for 20 ns;
53             assert(Q_tb_out = '0') report "Error in q" severity failure;
54
55             J_tb_in ≤ '1';
56             K_tb_in ≤ '1';
57             rst_n_tb ≤ '1';
58             wait for 20 ns;
59             assert(Q_tb_out = '0') report "Error in q" severity failure;

```




```

60     finish;
61     end process;
62 end architecture;

```

**Listing 4:** Testbench for JK flip-flop

The results of the simulation are shown in the [fig. 2](#) following.



**Figure 2:** Waveform of JK flip-flop

For the first 5ns, the Q output is Undefined, because all the inputs are set to 0 and the clock is in a falling edge which sets the flip-flop to a no-change state (default value of std\_logic is 'U').

### 3 Random Access Memory (RAM)

There are two types of memories that are used in digital systems: random access memory (RAM) and read only memory (ROM). A RAM stores new information for later use. The process of storing new information into memory is referred to as a memory write operation. The process of transferring the stored information out of memory is referred to as a memory read operation. RAM can perform both write and read operations.

The RAM implemented, is a 16 to 4. This means, 16 addresses to store 4 bits long data. An enable signal is used, to control the RAM. If enable is true, the system is operational, if not, the system is in an idle state. Besides that, a rw (read = 0, write = 1) is expected, to determine if the system is in a read or write state. Additionally, a vector of an address is expected, to determine which address to read or write. For the data input and output, a vector of 4 bits of inout is used, to combine in one. The sensitivity list only includes the clock signal.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity ram is
6      port(
7          clk, enable, rw: in std_logic;
8          address: in std_logic_vector (3 downto 0);
9          data: inout std_logic_vector (3 downto 0)
10     );
11 end ram;
12
13 architecture behavior of ram is
14     type ram_t is array (0 to 15) of std_logic_vector(3 downto 0);
15     signal tmp_ram: ram_t;
16 begin
17     process (clk)
18     begin
19         if (clk = '1' and clk'event) then
20             if enable = '1' then
21                 if rw = '1' then
22                     tmp_ram(to_integer(unsigned(address))) ≤ data;
23                 elsif rw = '0' then
24                     data ≤ tmp_ram(to_integer(unsigned(address)));
25                 else
26                     data ≤ (data'range ⇒ 'Z');
27                 end if;

```

```

28     end if;
29     end if;
30     end process;
31 end architecture;

```

#### Listing 5: RAM

The test bench for the RAM is shown in listing 6. The clock period is 1ns and the raise and fall time is 0.25ns.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use std.env.finish;
4
5  entity tb_ram is
6  end tb_ram;
7
8  architecture behavior of tb_ram is
9      signal clk: std_logic;
10     signal enable: std_logic;
11     signal rw: std_logic;
12     signal address: std_logic_vector (3 downto 0);
13     signal data: std_logic_vector (3 downto 0);
14     constant clk_period: time := 1 ns;
15 begin
16     RAM0: entity work.ram port map (clk, enable, rw, address, data);
17
18     clock : process
19     begin
20         for i in 1 to 20 loop
21             clk ≤ '1';
22             wait for clk_period/4;
23             clk ≤ '0';
24             wait for clk_period/4;
25         end loop;
26     finish;
27     end process;
28
29     process
30     begin
31         wait for clk_period;
32         enable ≤ '1';
33         wait for clk_period;
34         rw ≤ '1';
35         address ≤ "0011";
36         data ≤ "1001";
37         wait for clk_period;
38         rw ≤ '0';
39         address ≤ "0001";
40         data ≤ "ZZZZ";
41         wait for clk_period;
42         address ≤ "0011";
43         data ≤ "ZZZZ";
44         wait for clk_period;
45         wait;
46     end process;
47 end architecture;

```

#### Listing 6: Test bench for the RAM

The simulation of the RAM circuit, are presented with the following waveform image.




Figure 3: Waveform of RAM

## 4 Sequential multiplier

Multiplication of binary numbers is performed in the same way as multiplication of decimal numbers. The multiplicand is multiplied by each bit of the multiplier, starting from the least significant bit. Each such multiplication forms a partial product. Successive partial products are shifted one position to the left. The final product is obtained from the sum of the partial products.

The algorithm for the multiplication, is summed up by the pseudocode following.

---

**Algorithm 1:** Sequential multiplier

---

**Data:**  $A, B$  same  $n$ -bits binary numbers

**Result:**  $A \times B$

- 1  $P \leftarrow 0$
  - 2 Load  $A, B$
  - 3 **while**  $B \neq 0$  **do**
  - 4     **if**  $b_0 = 1$  **then**
  - 5          $P \leftarrow P + A$
  - 6     left shift  $A$
  - 7     right shift  $B$
- 

The digital circuit for the sequential multiplier, implements the procedure shown from before and the implementation is as follows below.




Figure 4: Circuit design of the sequential multiplier

Both the circuit and VHDL implementations, are developed in such way to calculate the multiplication of two  $n$ -bits numbers and not fixed length numbers.

The FSM following, shows the states that the sequential multiplier goes through to calculate the multiplication of two  $n$ -bits numbers. The states are represented by the circles and the arrows

represent the transitions between the states. The FSM instructs to store the value of the partial product, in the p register. Finally, the p register is also the output of the sequential multiplier.




Figure 5: Fine state machine

The sequential multiplier system, uses VHDL implementations of  $n$ -bits register,  $n$ -bits parallel access shift register and adder/subtractor circuits. For space-saving reasons, the VHDL implementations are not shown here, but can be found in the appendix.

```

1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  entity multiplier is
5      generic (N: INTEGER := 4);
6      port (
7          clk, reset, s: in std_logic;
8          dA, dB: in std_logic_vector (N-1 downto 0);
9          P: out std_logic_vector (2*N-1 downto 0);
10         done: out std_logic
11     );
12 end multiplier;
13
14 architecture Behavioral of multiplier is
15     component parallel_shift_reg
16         generic (
17             N: INTEGER := 4;
18             DIR: STRING := "LEFT"
19         );
20         port (
21             clk, reset: in std_logic;
22             din, E, s_l: in std_logic; --din: shiftn input
23             D: in std_logic_vector (N-1 downto 0);
24             Q: out std_logic_vector (N-1 downto 0);
25             shift_out: out std_logic
26         );
27     end component;
28
29     component add_sub
30         generic (N: INTEGER := 4);
31         port(
32             addsub: in std_logic;
33             x, y: in std_logic_vector (N-1 downto 0);
34             s: out std_logic_vector (N-1 downto 0);
35             overflow, cout: out std_logic
36         );
37     end component;
38
39     component n_register
40         generic (N: INTEGER := 4);
41         port (
42             clk, reset: in std_logic;
43             E, clear: in std_logic; -- clear: Synchronous clear
44             D: in std_logic_vector (N-1 downto 0);

```

```

45     Q: out std_logic_vector (N-1 downto 0)
46 );
47 end component;
48
49 type state is (S1, S2, S3);
50 signal y: state;
51
52 signal L, E, EP, sclrP, z: std_logic;
53 signal B: std_logic_vector (N-1 downto 0);
54 signal A, dAx, dP, Pt: std_logic_vector (2*N -1 downto 0);
55
56 begin
57 nA: parallel_shift_reg generic map (N => 2*N, DIR => "LEFT")
58   port map (
59     clk => clk, reset => reset, din => '0',
60     s_l => L, E => E, D => dAx, Q => A
61   );
62 dAx (2*N-1 downto N) <= (others => '0');
63 dAx (N-1 downto 0) <= dA;
64
65 nB: parallel_shift_reg generic map (N => N, DIR => "RIGHT")
66   port map (
67     clk => clk, reset => reset, din => '0',
68     s_l => L, E => E, D => dB, Q => B
69   );
70
71 -- n-bit NOR gate:
72 process (B)
73   variable result_or: std_logic;
74 begin
75   result_or := '0';
76   for i in B'range loop -- 'range: iterates through all bits in 'A'
77     result_or := result_or or B(i);
78   end loop;
79   z <= not (result_or);
80 end process;
81
82 --Adder:
83 ga: add_sub generic map (N => 2*N)
84   port map (addsub => '0', x => A, y => Pt, s => dP);
85
86 --Register P:
87 nP: n_register generic map (N => 2*N)
88   port map (
89     clk => clk, reset => reset, E => EP,
90     clear => sclrP, D => dP, Q => Pt
91   );
92 P <= Pt;
93
94 --FSM:
95 Transitions: process (reset, clk, s, z, B(0))
96 begin
97   if reset = '0' then -- asynchronous signal
98     y <= S1; -- if reset asserted, go to initial state: S1
99   elsif (clk'event and clk = '1') then
100     case y is
101       when S1 => if s = '1' then y <= S2; else y <= S1; end if;
102       when S2 => if z = '1' then y <= S3; else y <= S2; end if;
103       when S3 => if s = '1' then y <= S3; else y <= S1; end if;
104     end case;
105   end if;
106 end process;
107
108 Outputs: process (y, s, z, B(0))
109 begin
110 --Initialization of output signals
111 sclrP <= '0'; EP <= '0'; L <= '0'; E <= '0'; done <= '0';

```

```

112 case y is
113   when S1 =>
114     sclrP <= '1'; EP <= '1';
115     if s = '1' then
116       L <= '1'; E <= '1';
117     end if;
118   when S2 =>
119     E <= '1';
120     if z = '0' then
121       if B(0) = '1' then EP <= '1'; end if;
122     end if;
123   when S3 =>
124     done <= '1';
125 end case;
126 end process;
127 end Behavioral;

```

**Listing 7:**  $n$ -bits multiplier

Point of interest of the above implementation [listing 7](#) is the finite state machine (FSM) that is used to control the system. The FSM is implemented using a case statement, and the sensitivity list includes the clock signal. It follows exactly the FSM diagram is shown in [fig. 5](#) and the logic presented in that figure.

The test bench for the multiplier is shown in [listing 8](#). It has to be pointed out that, in worst case scenario, the multiplication of two  $n$ -bits numbers, needs  $n + 1$  clock cycles to complete. For that reason, on each multiplication the test bench waits  $n + 3$  clock cycles, to ensure that the multiplication is completed.

```

1  library ieee;
2  use ieee.std_logic_1164.ALL;
3  use ieee.std_logic_arith.all;
4  use std.env.finish;
5
6  entity tb_multiplier is
7    generic (N: INTEGER := 4);
8  end tb_multiplier;
9
10 architecture behavior of tb_multiplier is
11   --Inputs
12   signal clk: std_logic := '0';
13   signal reset: std_logic := '0';
14   signal s: std_logic := '0';
15   signal dA: std_logic_vector (N-1 downto 0) := (others => '0');
16   signal dB: std_logic_vector (N-1 downto 0) := (others => '0');
17
18   --Outputs
19   signal P: std_logic_vector (2*N - 1 downto 0);
20   signal done: std_logic;
21
22   -- Clock period definitions
23   constant clock_period: time := 10 ns;
24 begin
25   -- Instantiate the Unit Under Test (UUT)
26   uut: entity work.multiplier port map (clk, reset, s, dA, dB, P, done);
27
28   -- Clock process definitions
29   clock_process: process
30   begin
31     clk <= '0';
32     wait for clock_period/2;
33     clk <= '1';
34     wait for clock_period/2;
35   end process;
36
37   -- Stimulus process
38   stimulus: process

```


```

39 begin
40   -- hold reset state for 100 ns.
41   wait for 100 ns; reset ≤ '1';
42
43   dA ≤ conv_std_logic_vector (13, N); -- "1101";
44   dB ≤ conv_std_logic_vector (11, N); -- "1011";
45   s ≤ '1'; wait for clock_period;
46   s ≤ '0';
47   wait for clock_period*(N+3); -- Multiplication takes at most N + 1 cycles
48
49   dA ≤ conv_std_logic_vector (9, N);    -- "1001";
50   dB ≤ conv_std_logic_vector (15, N);   -- "1111";
51   s ≤ '1'; wait for clock_period;
52   s ≤ '0';
53   wait for clock_period*(N+3);
54   finish;
55 end process;
56 end architecture;

```

**Listing 8:** Test bench for the  $n$ -bits multiplier

The test bench in [listing 8](#), generates the waveform shown in [fig. 6](#). Before the end result is calculated, in the p register the partial multiplication result is stored.



**Figure 6:** Waveform of the multiplier

## Part II

# Combinatorial circuits

A combinational circuit consists of an interconnection of logic gates. Combinational logic gates react to the values of the signals at their inputs and produce the value of the output signal, transforming binary information from the given input data to a required output data.

## 5 LED Counter

The concept of a counter is quite simple: it is a circuit that counts. In this particular case, the counter is a 4-bit counter, which means that it can count from 0 to 15.

A LED counter is a combination of a LED decoder and a counter (see [Appendix A.1](#)). The LED Counter takes in the clock and reset inputs from the user and outputs the 7-bit value to be displayed on the LEDs. The LED Counter is a hierarchical design, where the LED Decoder and the Counter are instantiated as components in the LED Counter.

This design is hierarchical; both the LED decoder and the counter are autonomous units connected to each other through another higher-level unit to create the final circuit.

The VHDL code for the LED Counter is shown here [listing 9](#). The LED Counter has a clock and reset input, which are used to control the counter. The counter value is then passed to the LED Decoder, which returns the 7-bit value to be displayed on the LEDs. The LED Counter also has a 4-bit output, which is the counter value.

```

1 library ieee;
2 use ieee.std_logic_1164.all;
3

```

```

4 entity led_counter is
5   port (
6     clk, reset: in std_logic;
7     Q_out: out std_logic_vector (6 downto 0)
8   );
9 end entity;
10
11 architecture behavior of led_counter is
12   signal counter_value : std_logic_vector (3 downto 0);
13 begin
14   LED: entity work.led_decoder port map (clk, counter_value, Q_out);
15   COUNT: entity work.counter port map (clk, reset, counter_value);
16 end architecture;

```

**Listing 9:** LED Counter

The test bench for the LED Counter is shown here [listing 10](#). The test bench instantiates the LED Counter and provides the clock and reset inputs. The test bench also displays the counter value on the console. The test bench is run for 100ns, which is enough time for the counter to reach its maximum value and reset back to 0.

```

1 library ieee;
2 use ieee.std_logic_1164.all;
3 use std.env.finish;
4
5 entity tb_led_counter is
6 end entity;
7
8 architecture behavior of tb_led_counter is
9   constant clk_period: time := 10 ns;
10
11   signal clk: std_logic := '0';
12   signal reset: std_logic := '0';
13   signal led: std_logic_vector(6 downto 0) := "0000000";
14 begin
15   UUT: entity work.led_counter port map(clk, reset, led);
16
17   clk_process : process
18   begin
19     clk <= '0';
20     wait for clk_period/2;
21     clk <= '1';
22     wait for clk_period/2;
23   end process;
24
25   stimulus : process
26   begin
27     reset <= '1';
28     wait for 100 ns;
29     reset <= '0';
30     wait for 100 ns;
31     reset <= '1';
32     wait for 50 ns;
33     reset <= '0';
34     wait for 450 ns;
35     finish;
36   end process;
37 end architecture;

```

**Listing 10:** Test bench for LED Counter

The waveform extracted from GTKWave is presented in [fig. 7](#).






Figure 7: Waveform of LED Counter

## 6 Multiplexer

A multiplexer is a combinational circuit that selects binary information from one of many input lines and directs it to a single output line. The selection of a particular input line is controlled by a set of selection lines. Normally, there are  $2^n$  input lines and  $n$  selection lines whose bit combinations determine which input is selected.

Here the  $n$  value is 2, so this means that the MUX implemented is a 4 to 1 circuit. The truth table for the multiplexer is shown in [table 3](#).

Table 3: Truth table for the 4 to 1 MUX

$S_0$	$S_1$	$Q_{out}$
0	0	A
0	1	B
1	0	C
1	1	D

The circuit design, of a 4 to 1 MUX using digital logic gates, is shown in [fig. 8](#).




Figure 8: Circuit design of MUX 4 to 1

The developed code seen in [listing 11](#), uses a vector to store the input values, and then uses the selection lines, also a vector, to select the output value. The sensitivity list includes the selection and the input vectors, that means if the values of those two vectors the output gets recalculated. A small but important detail is that the output gets the value Uninitialized ('X') as a default value in the case statement, this is to avoid latches.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3
4  entity mux_4to1 is
5      port (
6          ABCD_in : in  std_logic_vector(3 downto 0);
7          Sel_in   : in  std_logic_vector(1 downto 0);
8          Q_out    : out std_logic
9      );
10 end entity;
```

```

11
12 architecture behavior of mux_4to1 is
13 begin
14     DUT: process(ABCD_in, Sel_in)
15     begin
16         case Sel_in is
17             when "00" => Q_out <= ABCD_in(0);
18             when "01" => Q_out <= ABCD_in(1);
19             when "10" => Q_out <= ABCD_in(2);
20             when "11" => Q_out <= ABCD_in(3);
21             when others => Q_out <= 'X';
22         end case;
23     end process;
24 end architecture;

```

**Listing 11:** MUX 4 to 1

The test bench for the listing 11 is shown in listing 12. To avoid long and repeated case statements, the test bench uses a for loop to iterate through all the possible combinations of the selection, input and output values. These values are stored in a separate file, and then read by the test bench. For this to work, the library STD is used.

```

1 library ieee;
2 use ieee.std_logic_1164.all;
3 use ieee.std_logic_textio.all;
4
5 library STD;
6 use STD.textio.all;
7
8 entity tb_mux_4to1 is
9 end entity;
10
11 architecture bench of tb_mux_4to1 is
12
13     component mux_4to1
14     port (
15         ABCD_in: in std_logic_vector(3 downto 0);
16         Sel_in: in std_logic_vector(1 downto 0);
17         Q_out: out std_logic
18     );
19 end component;
20
21 signal ACBD_tb_in: std_logic_vector(3 downto 0);
22 signal Sel_tb_in: std_logic_vector(1 downto 0);
23 signal Q_tb_in: std_logic;
24
25 begin
26     DUT: mux_4to1 port map(ABCD_in => ACBD_tb_in, Sel_in => Sel_tb_in, Q_out => Q_tb_in);
27     process
28         file Fin : text open read_mode is "mux_4to1_tests.txt";
29
30         variable current_read_line : line;
31         variable current_read_field1 : std_logic_vector(0 to 3);
32         variable current_read_field2 : std_logic_vector(0 to 1);
33         variable current_read_field3 : std_logic;
34
35     begin
36         while (not endFile(Fin)) loop
37
38             readline(Fin, current_read_line);
39             read(current_read_line, current_read_field1);
40             read(current_read_line, current_read_field2);
41             read(current_read_line, current_read_field3);
42
43             ACBD_tb_in <= current_read_field1;
44             Sel_tb_in <= current_read_field2;
45             wait for 50 ns;

```


```

46
47     assert(Q_tb_in = current_read_field3);
48
49     end loop;
50     wait;
51     end process;
52 end architecture;

```

**Listing 12:** Testbench for MUX 4 to 1

The simulation of the [listing 11](#) creates the waveform shown in [fig. 9](#).



**Figure 9:** Waveform of MUX 4 to 1

Which is the expected result, because the output value is the same as the input value that is selected by the selection lines.

## 7 Full adder

A full adder is a combinational circuit that forms the arithmetic sum of three bits. It consists of three inputs and two outputs. Two of the input variables, denoted by  $x$  and  $y$ , represent the two significant bits to be added. The third input,  $z$ , represents the carry from the previous stage. Two outputs are necessary because the arithmetic sum of three binary digits ranges in value from 0 to 3, and binary representation of 2 or 3 needs two bits. The two outputs are designated by the symbols  $S$  for sum and  $C$  for carry. The binary variable  $S$  gives the value of the least significant bit of the sum. The binary variable  $C$  gives the output carry formed by adding the input carry and the bits of the words. The truth table of the full adder is listed in [table 4](#).

**Table 4:** Truth table for the 3 bit full adder

$x$	$y$	$c_{in}$	$C_{out}$	$S$
0	0	0	0	0
0	0	1	0	1
0	1	0	0	1
0	1	1	1	0
1	0	0	0	0
1	0	1	0	0
1	1	0	1	0
1	1	1	1	1

The circuit design, of a full adder using digital logic gates, is shown in [fig. 10](#).




Figure 10: Circuit design of a full adder

The VHDL code for the full adder is shown in listing 13. This implementation is quite simple and follows the basic description from the design shown in fig. 10. The sensitivity list includes all the inputs, so if any of them changes the output gets recalculated.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3
4  entity full_adder is
5      port (
6          x_in, y_in, c_in : in  std_logic;
7          s_out, c_out      : out std_logic
8      );
9  end entity;
10
11 architecture behavior of full_adder is
12 begin
13     DUT: process(x_in, y_in, c_in)
14     begin
15         s_out ≤ x_in xor y_in xor c_in;
16         c_out ≤ (x_in and y_in) or (x_in and c_in) or (y_in and c_in);
17     end process;
18 end architecture;

```

Listing 13: Full adder

For the same reasons as before mentioned, the test bench for the full adder uses a for loop to iterate through all the possible combinations of the inputs and outputs. If an unexpected value is presented, to assert instruction used, will underline the error. The test bench implantation is shown in below.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.std_logic_textio.all;
4
5  library STD;
6  use STD.textio.all;
7
8  entity tb_full_adder is
9  end entity;
10
11 architecture bench of tb_full_adder is
12     component full_adder
13     port (
14         x_in, y_in, c_in : in  std_logic;
15         s_out, c_out      : out std_logic

```

```

16     );
17 end component;
18
19 signal x_tb_in  : std_logic;
20 signal y_tb_in  : std_logic;
21 signal c_tb_in  : std_logic;
22 signal s_tb_out : std_logic;
23 signal c_tb_out : std_logic;
24
25 begin
26   DUT: full_adder port map(x_in => x_tb_in, y_in => y_tb_in, c_in => c_tb_in, s_out =>
27     ↪ s_tb_out, c_out => c_tb_out);
28   process
29
30     file Fin : text open read_mode is "full_adder_input.txt";
31
32     variable current_read_line  : line;
33     variable current_read_field1 : std_logic;
34     variable current_read_field2 : std_logic;
35     variable current_read_field3 : std_logic;
36     variable current_read_field4 : std_logic;
37     variable current_read_field5 : std_logic;
38
39     begin
40       while (not endFile(Fin)) loop
41
42         readline(Fin, current_read_line);
43         read(current_read_line, current_read_field1);
44         read(current_read_line, current_read_field2);
45         read(current_read_line, current_read_field3);
46         read(current_read_line, current_read_field4);
47         read(current_read_line, current_read_field5);
48
49         x_tb_in <= current_read_field1;
50         y_tb_in <= current_read_field2;
51         c_tb_in <= current_read_field3;
52
53         wait for 50 ns;
54
55         assert(c_tb_out = current_read_field4) report "Error in carry" severity failure;
56         assert(s_tb_out = current_read_field5) report "Error in sum" severity failure;
57       end loop;
58       wait;
59     end process;
60 end architecture;

```

Listing 14: Test bench for the full adder

## 8 Comparator

A (magnitude) comparator is a combinational circuit that compares two numbers A and B and determines their relative magnitudes. The outcome of the comparison is specified by a variable which indicates, if the numbers are equal or not.

The VHDL implementation following in listing 15, uses generic value for the size of the numbers, this means that the comparator can be used for any size of numbers.

```

1 library ieee;
2 use ieee.std_logic_1164.all;
3
4 entity comparator is
5   generic(N: integer:= 4);
6   port (
7     x_in, y_in: in std_logic_vector(N-1 downto 0);

```

```

8     is_eq: out std_logic
9 );
10 end entity;
11
12 architecture behavior of comparator is
13 begin
14     DUT: process(x_in, y_in)
15     begin
16         is_eq ≤ '1' when (x_in = y_in)
17         else '0';
18     end process;
19 end architecture;

```

**Listing 15:** Comparator

For the testing of the comparator, the  $N$  value was set to 4, i.e. 4 bits numbers. The test bench for the comparator is shown in [listing 16](#).


```

1 library ieee;
2 use ieee.std_logic_1164.all;
3 use std.env.finish;
4
5 entity tb_comparator is
6 end tb_comparator;
7
8 architecture behavior of tb_comparator is
9     signal x_tb_in, y_tb_in: std_logic_vector(3 downto 0) := "0000";
10    signal is_eq_tb: std_logic;
11    constant clk: time := 1 ns;
12 begin
13    DUT: entity work.comparator port map (x_tb_in, y_tb_in, is_eq_tb);
14    process
15    begin
16        wait for clk;
17        x_tb_in ≤ "0001";
18        y_tb_in ≤ "0000";
19        wait for clk;
20        x_tb_in ≤ "0110";
21        y_tb_in ≤ "0110";
22        wait for clk;
23        wait;
24    finish;
25    end process;
26 end architecture;

```

**Listing 16:** Test bench for Comparator

The simulation results of the above test bench are summed up in the following waveform image.



**Figure 11:** Waveform of Comparator

## 9 Encoder

An encoder is a digital circuit that performs the inverse operation of a decoder. An encoder has  $2^n$  input lines and  $n$  output lines. The output lines, as an aggregate, generate the binary code corresponding to the input value.

In this implementation, the encoder is a 16 to 4 circuit. The table for how the systems switches between the given values is shown below [table 5](#).

**Table 5:** Truth table for the 16 to 4 encoder

Input (16-bits)	Output (4-bits)
0000000000000001	0000
0000000000000010	0001
0000000000000100	0010
0000000000001000	0011
0000000000010000	0100
0000000000100000	0101
0000000001000000	0110
0000000010000000	0111
0000000100000000	1000
0000001000000000	1001
0000010000000000	1010
0000100000000000	1011
0001000000000000	1100
0010000000000000	1101
0100000000000000	1110
1000000000000000	1111

The implemented encoder, shown in [listing 17](#), reads the input vector and then uses a case statement to select the output value. The sensitivity list only consists of the input vector. A default value of Undefined ('XXXX') is also used to avoid latches.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3
4  entity encoder_16to4 is
5      port (
6          D_in: in std_logic_vector(15 downto 0);
7          Q_out: out std_logic_vector(3 downto 0)
8      );
9  end entity;
10
11 architecture behavior of encoder_16to4 is
12 begin
13     DUT: process(D_in)
14     begin
15         case D_in is
16             when "0000000000000001" => Q_out <= "0000";
17             when "0000000000000010" => Q_out <= "0001";
18             when "0000000000000100" => Q_out <= "0010";
19             when "0000000000001000" => Q_out <= "0011";
20             when "0000000000010000" => Q_out <= "0100";
21             when "0000000000100000" => Q_out <= "0101";
22             when "0000000001000000" => Q_out <= "0110";
23             when "0000000010000000" => Q_out <= "0111";
24             when "0000000100000000" => Q_out <= "1000";
25             when "0000001000000000" => Q_out <= "1001";
26             when "0000010000000000" => Q_out <= "1010";
27             when "0000100000000000" => Q_out <= "1011";
28             when "0001000000000000" => Q_out <= "1100";
29             when "0010000000000000" => Q_out <= "1101";
30             when "0100000000000000" => Q_out <= "1110";
31             when "1000000000000000" => Q_out <= "1111";
32             when others => Q_out <= "XXXX";
33         end case;
34     end process;
35 end architecture;

```

**Listing 17:** Encoder 16 to 4


For the testing of the encoder, a text file was used to store the input and output values and like shown before, the contents of the file were read and compared with the values that the encoder produced.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.std_logic_textio.all;
4
5  library STD;
6  use STD.textio.all;
7
8  entity tb_encoder_16to4 is
9  end entity;
10
11 architecture behavior of tb_encoder_16to4 is
12     signal D_tb_in: std_logic_vector(15 downto 0);
13     signal Q_tb_out: std_logic_vector(3 downto 0);
14 begin
15     DUT: entity work.encoder_16to4 port map (D_tb_in, Q_tb_out);
16     process
17         file f_in: text open read_mode is "encoder_16to4_test.txt";
18
19         variable current_read_line: line;
20         variable current_read_field1: std_logic_vector(0 to 15);
21         variable current_read_field2: std_logic_vector(0 to 3);
22     begin
23         while (not endFile(f_in)) loop
24             readline(f_in, current_read_line);
25             read(current_read_line, current_read_field1);
26             read(current_read_line, current_read_field2);
27
28             D_tb_in ≤ current_read_field1;
29             wait for 50 ns;
30
31             assert(Q_tb_out = current_read_field2);
32         end loop;
33         wait;
34     end process;
35 end architecture;
```

**Listing 18:** Encoder 16 to 4

The simulation results of the test bench in listing 18, are shown in fig. 12. The easy-to-understand functionality of the circuit, makes it possible to verify the results by looking in the waveform.



**Figure 12:** Waveform of the encoder



## A Supplementary code for LED counter & multiplier

### A.1 LED counter

#### A.1.1 LED Decoder

The LED Decoder is a combination of LEDs that should display the value of the decoder at any given moment. The decoder takes in 4-bits at its input and outputs 7 bits. Essentially, the LED will represent the hexadecimal value of the 4-bits as shown in the table below.

**Table 6:** LED Decoder

Binary value (4-bits)	LED value
0000	0
0001	1
0010	2
0011	3
0100	4
0101	5
0110	6
0111	7
1000	8
1001	9
1010	A
1011	B
1100	C
1101	D
1110	E
1111	F

The VHDL code for the LED Decoder is shown here [listing 19](#).

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.std_logic_unsigned.all;
4  use ieee.std_logic_arith.ALL;
5
6  entity led_decoder is
7      port (
8          clk: in std_logic;
9          d: in std_logic_vector (3 downto 0);
10         s: out std_logic_vector (6 downto 0)
11     );
12 end entity;
13
14 architecture behavior of led_decoder is
15 begin
16     DUT: process(clk, d)
17     begin
18         if rising_edge(clk) then
19             s <= "1110111" when d="0000" else
20                 "0010010" when d="0001" else
21                 "1011101" when d="0010" else
22                 "1011011" when d="0011" else
23                 "0111010" when d="0100" else
24                 "1101011" when d="0101" else
25                 "1101111" when d="0110" else
26                 "1010010" when d="0111" else
27                 "1111111" when d="1000" else
28                 "1111011" when d="1001" else

```

```

29     "1111110" when d="1010" else
30     "0101111" when d="1011" else
31     "1100101" when d="1100" else
32     "0011111" when d="1101" else
33     "1101101" when d="1110" else
34     "1101100";
35 end if;
36 end process;
37 end architecture;

```

Listing 19: LED Decoder

### A.1.2 Counter

The counter implemented, is a 4-bit counter that counts from 0 to 15. The increment of the counter is a synchronous manner, i.e. the counter value is updated only when the clock signal is high. When the counter reaches its maximum value (15), it resets back to 0 (wrap around).

The VHDL code for the counter is shown here listing 20. The reset input is used to reset the counter back to 0. Besides that, the counter also has a clock input, which is used to increment the counter value. The counter value returned is a 4-bit value, using a vector.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.std_logic_unsigned.all;
4
5  entity counter is
6      port (
7          clk, reset: in std_logic;
8          count: out std_logic_vector (3 downto 0)
9      );
10 end entity;
11
12 architecture behavior of counter is
13 begin
14     process(clk, reset)
15     begin
16         if reset = '1' then
17             count ≤ "0000";
18         elsif rising_edge(clk) then
19             if count = "1001" then
20                 count ≤ "0000";
21             else
22                 count ≤ count + 1;
23             end if;
24         end if;
25     end process;
26 end architecture;

```

Listing 20: Counter

## A.2 Multiplier

### A.2.1 Adder/Subtractor

```

1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  entity add_sub is
5      generic (N: INTEGER:= 4);
6      port(
7          addsub: in std_logic; -- addsub = 0 (add), addsub = 1 (sub)
8          x, y: in std_logic_vector (N-1 downto 0);
9          s: out std_logic_vector (N-1 downto 0);

```

```

10     overflow: out std_logic;
11     cout: out std_logic);
12 end add_sub;
13
14 architecture structure of add_sub is
15     component full_adder
16     port(
17         x_in, y_in, c_in : in std_logic;
18         s_out, c_out: out std_logic
19     );
20     end component;
21
22     signal c: std_logic_vector (N downto 0);
23     signal yx: std_logic_vector (N-1 downto 0);
24 begin
25     c(0) ≤ addsub;
26     cout ≤ c(N);
27     overflow ≤ c(N) xor c(N-1);
28
29     gi: for i in 0 to N-1 generate
30         yx(i) ≤ y(i) xor addsub;
31         fi: full_adder port map (
32             c_in ⇒ c(i), x_in ⇒ x(i), y_in ⇒ yx(i), s_out ⇒ s(i),
33             c_out ⇒ c(i+1)
34         );
35     end generate;
36 end structure;

```

Listing 21:  $n$ -bits adder/subtractor

## A.2.2 Register

```

1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  -- N-bit Register
5  -- E = '1', clear = '0' --> Input data 'D' is copied on Q
6  -- E = '1', clear = '1' --> Q is cleared (0)
7  entity n_register is
8      generic (N: INTEGER:= 4);
9      port (
10         clk, reset : in std_logic;
11         E, clear : in std_logic; -- clear: Synchronous clear
12         D : in std_logic_vector (N-1 downto 0);
13         Q : out std_logic_vector (N-1 downto 0));
14 end n_register;
15
16 architecture Behavioral of n_register is
17     signal Qt : std_logic_vector (N-1 downto 0);
18 begin
19     process (reset, clk)
20     begin
21         if reset = '0' then
22             Qt ≤ (others ⇒ '0');
23         elsif (clk'event and clk = '1') then
24             if E = '1' then
25                 if clear = '1' then
26                     Qt ≤ (others ⇒ '0');
27                 else
28                     Qt ≤ D;
29                 end if;
30             end if;
31         end if;
32     end process;
33     Q ≤ Qt;

```

```

34 end Behavioral;
35

```

Listing 22:  $n$ -bits register

## A.2.3 Parallel Shift Register

```

1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  -- N-bit Parallel access Shift register:
5  -- DIR = "LEFT" --> Shift to the left: from LSB to MSB
6  -- DIR = "RIGHT" --> Shift to the right: from MSB to LSB
7  -- s_l = 0 --> Shift operation (MSB to LSB)
8  -- s_l = 1 -> Parallel load
9  entity parallel_shift_reg is
10     generic (
11         N : INTEGER := 4;
12         DIR : STRING := "LEFT"
13     );
14
15     port (
16         clk, reset : in std_logic;
17         din, E, s_l : in std_logic; -- din: shiftin input
18         D : in std_logic_vector (N-1 downto 0);
19         Q : out std_logic_vector (N-1 downto 0);
20         shift_out : out std_logic
21     );
22 end parallel_shift_reg;
23
24 architecture Behavioral of parallel_shift_reg is
25     signal Qt: std_logic_vector (N-1 downto 0);
26 begin
27
28     a0: assert (DIR = "LEFT" or DIR = "RIGHT")
29         report "DIR can only be LEFT or RIGHT" severity error;
30
31     process (reset, clk)
32     begin
33         if reset = '0' then
34             Qt ≤ (others ⇒ '0');
35         elsif (clk'event and clk = '1') then
36             if E = '1' then
37                 if s_l = '1' then
38                     Qt ≤ D;
39                 else
40                     if DIR = "LEFT" then
41                         Qt(0) ≤ din;
42                         for i in 1 to N-1 loop
43                             Qt(i) ≤ Qt(i-1);
44                         end loop;
45                     elsif DIR = "RIGHT" then
46                         Qt(N-1) ≤ din;
47                         for i in 0 to N-2 loop
48                             Qt(i) ≤ Qt(i+1);
49                         end loop;
50                     end if;
51                 end if;
52             end if;
53         end if;
54
55     end process;
56
57     Q ≤ Qt;
58
59     gl: if DIR = "LEFT" generate

```

```
60     shift_out ≤ Qt(N-1);  
61   end generate;  
62   gr: if DIR = "RIGHT" generate  
63     shift_out ≤ Qt(0);  
64   end generate;  
65 end Behavioral;
```

**Listing 23:**  $n$ -bits parallel shift register

## B Tools used for digital design and VHDL compilation

There are a lot of development environments suites for analyzing, compiling, simulating and synthesizing VHDL code. Those products come with a hefty price for licensing and the proprietary. For this project, a vocal point was to use free and open source software.

For the analyzing, compiling and simulating of the VHDL code, the open source software GHDL was used. Unlike some other simulators, GHDL is a compiler: it directly translates a VHDL file to machine code, without using an intermediary language such as C/C++.

If synthesizing was needed, the open source software Yosys is an excellent tool for that.

To validate the behavior of the designs, the test benches simulations were exported as vcd files and then imported into GTKWave, which is an external waveform viewer.

The design of some circuits with logic gates, was done using the open source software Logisim-Evolution. This software is a digital logic designer and simulator. It allows the user to design and simulate digital logic circuits.

Finally, the IDE of choice for this project was VSCodium with the extension of ghdl-ls, that acts as a language server for VHDL. This extension provides syntax highlighting, code completion, linting and other features. Additionally, for each design, a Makefile was created to automate the process of compiling and simulating the VHDL code.

### How to use GHDL and GTKWave

The GHDL framework is available through various package managers. The most common way to install GHDL is by building yourself from source. The instructions for building GHDL from source are available [here](#). For the GTKWave viewer, visit their official website and follow from there the instructions for your operating system.

For each design, a source file of the particular design is needed, which can consists of one or more VHDL files in a hierarchical way. Furthermore, a test bench file is needed to simulate the design. The test bench file is usually named with the prefix `tb_` followed by the name of the design.

For each design the following instructions have to be used:

- Analyze the source file(s):

```
ghdl -a -std=08 -ieee=synopsys <design>.vhd
```

- Analyze the testbench file(s):

```
ghdl -a -std=08 -ieee=synopsys tb_<design>.vhd
```

- Generate executable file:

```
ghdl -e -std=08 -ieee=synopsys tb_<design>.vhd
```

- Run the simulation:

```
ghdl -r -std=08 -ieee=synopsys tb_<design>.vhd
      -vcd=tb_<design>.vcd
```

- View the waveform:

```
gtkwave tb_<design>.vcd
```

Where `<design>` is the name of the design. If all tests are passed, then the no errors will be shown on the terminal. Alternatively, for each design the Makefile can be used to automate the process.

## References

- [1] M. M. Mano and M. D. Ciletti, *Digital Design (4th Edition)*. USA: Prentice-Hall, Inc., 2006.
- [2] S. Brown and Z. Vranesic, *Fundamentals of Digital Logic with VHDL Design with CD-ROM*. USA: McGraw-Hill, Inc., 3 ed., 2008.
- [3] D. Llamocca, “Unit 7 - introduction to digital system design.” <https://www.secs.oakland.edu/llamocca/>, 2023.
- [4] V. Angelov, “Vhdl vorlesung ss2009.” [https://www.physi.uni-heidelberg.de/~angelov/VHDL/VHDL\\_SS09\\_Teil05.pdf](https://www.physi.uni-heidelberg.de/~angelov/VHDL/VHDL_SS09_Teil05.pdf), 2009.